ISSN 0704-3716

Canadian Translation of Fisheries and Aquatic Sciences

No. 5103

The BASIC program to analyse the polymodal frequency distribution into normal distributions with Marqualdt's method

T. Akamine

Original title: Marqualdt-ho ni yoru Polymodal-na Dosu Bunpu o Seiki Bunpu e Bunkai suru BASIC puroguramu

In: Bull. Jpan. Sea Reg. Fish. Res. Lab. (34): 53-60, 1984

Original language: Japanese

Available from:

Canada Institute for Scientific and Technical Information
National Research Council
Ottawa, Ontario, Canada K1A 0S2

1984

12 typescript pages

DEPARTMENT OF THE SECRETARY OF STATE
TRANSLATION BUREAU

MULTILINGUAL SERVICES DIVISION

CANADA

SECRÉTARIAT D'ÉTAT
BUREAU DES TRADUCTIONS

DIVISION DES SERVICES MULTILINGUES

CTFAS 5103

TRANSLATED FROM — TRADUCTION DE	INTO — EN
Japanese	English

AUTHOR — AUTEUR

Tatsuro AKAMINE

TITLE IN ENGLISH — TITRE ANGLAIS

The BASIC Program to Analyse the Polymodal Frequency Distribution into Normal Distributions with Marqualdt's Method

TITLE IN FOREIGN LANGUAGE (TRANSLITERATE FOREIGN CHARACTERS)
TITRE EN LANGUE ÉTRANGÈRE (TRANSCRIRE EN CARACTÈRES ROMAINS)

Marqualdt-ho ni yoru Polymodal-na Dosu Bunpu o Seiki Bunpu e Bunkai suru BASIC puroguramu

REFERENCE IN FOREIGN LANGUAGE (NAME OF BOOK OR PUBLICATION) IN FULL. TRANSLITERATE FOREIGN CHARACTERS.
RÉFÉRENCE EN LANGUE ÉTRANGÈRE (NOM DU LIVRE OU PUBLICATION), AU COMPLET, TRANSCRIRE EN CARACTÈRES ROMAINS.

Nipponkai-ku Suisan Kenkyusho Hokoku

REFERENCE IN ENGLISH — RÉFÉRENCE EN ANGLAIS

Bulletin of the Japan Sea Regional Fisheries Research Laboratory

PUBLISHER — ÉDITEUR	DATE OF PUBLICATION / DATE DE PUBLICATION			PAGE NUMBERS IN ORIGINAL / NUMÉROS DES PAGES DANS L'ORIGINAL
–	YEAR / ANNÉE	VOLUME	ISSUE NO. / NUMÉRO	53 – 60
PLACE OF PUBLICATION / LIEU DE PUBLICATION				NUMBER OF TYPED PAGES / NOMBRE DE PAGES DACTYLOGRAPHIÉES
Japan	1984	–	34	12

REQUESTING DEPARTMENT / MINISTÈRE-CLIENT: DFO

TRANSLATION BUREAU NO. / NOTRE DOSSIER N° 1655822

BRANCH OR DIVISION / DIRECTION OU DIVISION: SIPB

TRANSLATOR (INITIALS) / TRADUCTEUR (INITIALES): PS

PERSON REQUESTING / DEMANDÉ PAR: A. T. Reid

OCT - 5 1984

YOUR NUMBER / VOTRE DOSSIER N°

DATE OF REQUEST / DATE DE LA DEMANDE: September 6, 1984

UNEDITED TRANSLATION

SOS-200-10-6 (REV. 2/68)
7530-21-029-5333

Client's No.—N° du client	Department — Ministère	Division/Branch — Division/Direction	City — Ville
	DFO	SIPB	Ottawa
Bureau No.—N° du bureau	**Language — Langue**	**Translator (Initials) — Traducteur (Initiales)**	
1655822	Japanese	PS	OCT - 5 1984

The BASIC Program to Analyse the Polymodal Frequency Distribution into Normal Distributions with Marqualdt's Method

by

Tatsuro AKAMINE

Abstract

The algorithm of this program is Marqualdt's method. Gauss' elimination method is used to solve the simultaneous linear equations. Each parameter is scaled during calculation for faster convergence. User inputs the data and initial values of the parameters. It is adequate for convergence to set $\lambda=10000$ or larger.

The method analysing the polymodal frequency distribution into normal distribution enables by plotting the frequencies in the medium values of the classes to perform a resolution to the regression curve method based on the least square. Akamine (1982) prepared the program of the Gauss-Seidel method. As the memory was reduced to the minimum for a small-size computer, there was a drawback of slow convergence.

The performance of small computers has risen remarkably in recent times and it became possible to prepare programs with more rapid convergence. The present program based on the Marqualdt method is rapid

[1] Japan Sea Regional Fisheries Research Laboratory, Suido-cho, Niigata 951, Japan.

in convergence and good in stability. The machine used is the PC 8801 (NEC), but if the memory is satisfactory, less advanced machines may be used.

Calculation Method

We will call F the given frequency distribution, m the number of classes, h the width of the classes, a the smallest class value and b the greatest value. If we call n the number of normal distributions to be analysed, then the sought formula f, the residual function d^2 are:

$$f=\sum_{i=1}^{n} K_i \cdot N(\mu_i, \sigma_i, x) \quad \cdots\cdots\cdots\cdots\cdots\cdots\cdots\cdots ①$$

$$N(\mu_i, \sigma_i, x)=\frac{1}{\sqrt{2\pi}\,\sigma_i}\exp\left\{-\frac{(x-\mu_i)^2}{2\sigma_i^2}\right\}$$

$$d^2=\sum_{x=a}^{b} d_x^2=\sum_{x}(F-f)^2 \quad \cdots\cdots\cdots\cdots\cdots ②$$

$$b \fallingdotseq a+(m-1)h$$

(Fig. 1).

From ① we have:

$$\begin{cases} \dfrac{\partial f}{\partial K_i}=N(\mu_i, \sigma_i, x) \\ \dfrac{\partial f}{\partial \mu_i}=K_i \cdot N(\mu_i, \sigma_i, x) \cdot \dfrac{x-\mu_i}{\sigma_i^2} \\ \dfrac{\partial f}{\partial \sigma_i}=K_i \cdot N(\mu_i, \sigma_i, x) \cdot \dfrac{(x-\mu_i)^2-\sigma_i^2}{\sigma_i^3} \end{cases}$$

1. The Gauss-Seidel and the Gauss-Newton methods

With a primary approximation of ① , we have:

$$\Delta f \fallingdotseq \sum_{i}^{n}\left\{\frac{\partial f}{\partial K_i}\Delta K_i+\frac{\partial f}{\partial \mu_i}\Delta \mu_i+\frac{\partial f}{\partial \sigma_i}\Delta \sigma_i\right\}$$

If rearranging here the parameters K_i, μ_i, σ_i we express by p. 54

α_i = (i = 1-3n), we obtain:

$$\Delta f \fallingdotseq \sum_{i}^{3n}\frac{\partial f}{\partial \alpha_i}\Delta \alpha_i$$

If we substitute this in ②, then:

$$d^2=\sum_{x}\left\{F-(f+\Delta f)\right\}^2$$

$$=\sum_{x}\left(d_x-\sum_{i}\frac{\partial f}{\partial \alpha_i}\Delta \alpha_i\right)^2$$

If we use here the least square method, then:

$$\frac{\partial d^2}{\partial \Delta\alpha_j} = -2\sum_x \left(d_x - \sum_i \frac{\partial f}{\partial \alpha_i}\Delta\alpha_i\right)\frac{\partial f}{\partial \alpha_j} = 0$$

From this we obtain the following simultaneous equation:

$$\left(\sum_x \frac{\partial f}{\partial \alpha_i}\,\frac{\partial f}{\partial \alpha_j}\right)\left(\Delta\alpha_i\right) = \left(\sum_x d_x \frac{\partial f}{\partial \alpha_i}\right) \cdots ③$$

This we call a normal equation and abbreviate as

$$A\,\Delta\,\alpha = b$$

This is solved by the Gauss-Newton method which seeks $\Delta\alpha_i$. In fact, however, the methods not using directly the normal equation are more precise and adopted by apparatus such as the large computers (Nakagawa-Koyanagi 1982). We used here the method solving the normal equation by the elimination method of Gauss. However, as the coefficient queue consists of symmetric positive fixed values, the computation is performed in an upper triangular queue (Togawa 1971).

As the parameters are moved one at a time in the Gauss-Seidel method, because of $\Delta\alpha_j = 0$ ($j \neq i$), ③ becomes:

$$\sum_x \left(\frac{\partial f}{\partial \alpha_i}\right)^2 \Delta\alpha_i = \sum_x \left(d_x \frac{\partial f}{\partial \alpha_i}\right)$$

and the solution is easy. When explained in the case of 2 variables, Fig. 2 is obtained and we see that the convergence is slow.

In this regard, with linear patterns a solution is obtained at once in the Gauss-Newton method as with regression straight lines. With non-linear patterns, repetition is necessary because of the error due to linear pattern approximation. However, convergence is extremely rapid in the vicinity of the solution. Farther away from the solution there is occurrence of vibration and dispersion.

In order to compare both methods, we analysed the data of porgy body length composition by Tanaka (1956). With the Gauss-Seidel method it takes 500 iterations to reach a perfect convergence. With the Gauss-Newton method, a perfect convergence is reached after 5 or less iterations. However in the latter method, there is vibration and dispersion with ordinary early values, and so for the early values we used values calculated 5 times by the Gauss-Seidel method.

2. Steepest descent method p. 55

Calling the parameter α_i, ② yields:

$$\frac{\partial d^2}{\partial \alpha_i} = -2\sum_x d_x \frac{\partial f}{\partial \alpha_i} \quad \cdots\cdots\cdots\cdots ④$$

If we assume:

$$g = \left(-\frac{\partial d^2}{\partial \alpha_i}\right)$$

then the vector g is the normal of the equal-height surface of d^2 and faces the direction of the valley bottom (Fig. 4). We call g the steepest descent vector. Thus we put

$$\Delta \alpha_i = \left(\sum_x d_x \frac{\partial f}{\partial \alpha_i}\right) k$$

and by taking a suitable k we repeat the calculation until the solution is reached. Generally one provides a suitable early value k_o and when $\Delta d^2 < 0$, there is acceleration by $k_{n+1} = k_n * 1.2$. When $\Delta d^2 \geqslant 0$, there is acceleration by $k_{n+1} = k_n/2$ and one proceeds anew (Ruckdeschel 1982).

If we take a method similar to the Gauss-Seidel method, then from p. 56

$$\Delta f \doteqdot \sum_i \frac{\partial f}{\partial \alpha_i} \Delta \alpha_i = \left\{\sum_i \frac{\partial f}{\partial \alpha_i}\left(\sum_x d_x \frac{\partial f}{\partial \alpha_i}\right)\right\} k$$

$$= \beta_x k$$

$$d^2 = \sum_x \left\{F - (f + \Delta f)\right\}^2 = \sum_x (d_x - \beta_x k)^2$$

$\frac{\partial d^2}{\partial k} = 0$ k is found through $k = \frac{\sum_x d_x \beta_x}{\sum_x \beta_x^2}$

However, no convergence takes place by this alone. The reason is that because of $\frac{\partial f}{\partial K_i} < \frac{\partial f}{\partial \mu_i}, \frac{\partial f}{\partial \sigma_i}$, K_1 does not change at all. So we performed scaling with $\alpha_i' = \frac{\alpha_i}{a_i}$. We obtain at this time

$$\frac{\partial f}{\partial \alpha_i'} = \frac{\partial f}{\partial \alpha_i} \frac{\partial \alpha_i}{\partial \alpha_i'} = a_i \cdot \frac{\partial f}{\partial \alpha_i}$$

$$\Delta \alpha_i = a_i \cdot \Delta \alpha_i'$$

We chose here α_i and used the value of α_i at this time. Consequently all parameters of every scale become 1 (Fig. 5). With this scaling there is a fair improvement of convergence. In simple cases, this will do but in this case the convergence is slow and not practical (Table 1). The reason is that although the steepest descent direction is locally the optimum direction, it is not necessarily a good general direction (Nakagawa, Koyanagi 1982). Thus the steepest descent method is stable but its characteristic is a slow convergence.

3. The Marqualdt method

The Marqualdt method combines the mutally opposite Gauss-Newton method and the steepest descent method. When comparing (3) and (4) we have b = g. Thus:

$$(A+\lambda I)\Delta\alpha=b$$

We pose I: unit sequence. When λ is large, $\lambda\ I\ \Delta\ \alpha = b$ is approximated and as $\Delta\alpha = (1/\lambda)b$, we have the steepest descent method. When λ is small $A\Delta\alpha = b$ is approximated and we have the Gauss-Newton method. Consequently, we take a large λ early value and as it grows smaller, the solution is reached. When in the Marqualdt method $\Delta d^2 < 0$, λ is decreased by $\lambda = \lambda/\nu$. When $\Delta d^2 \geqslant 0$, λ is increased by $\lambda = \lambda\cdot\nu$ and the procedure is repeated. Generally $\nu = 2$ (Shimazu 1979). There is a method to obtain Δd^2 through primary approximation by

$$\Delta d^2 \doteqdot -2\sum_x d_x\left(\sum_i \frac{\partial f}{\partial \alpha_i}\Delta\alpha_i\right)$$

but vibration and dispersion occurred. The reason seems to be that the non-rectilinear pattern characteristic is strong and the error large with a primary approximation. Here we find Δd^2 directly by $\Delta d^2 = d^2_{n+1} - d^2_n$ and use this for the determination.

Example of application

Using a PC 8801, we analysed the data of Tanaka. The results are shown in Table 1. When λ is taken above 1,000, the convergence is extremely good. When λ is taken at 100 or lower, $\lambda \to \infty$ after the 2nd time and a stop p. 57
is reached. The reason seems to be that because λ is too small, the contribution of λ is small and with the first computation, there is a change due to the Gauss-Newton method and a stoppage point is entered.

In the Marqualdt method one must take a large λ and start in a condition close to the steepest descent method. In this case λ decreases regularly and one converges on the solution. When to the contrary λ increases regularly, the solution is neared or λ becomes too small and a stop point or another very small point is entered so that one has to stop. In this program, λ is continuous and one stops when a 10-fold increase takes place.

As for the value of λ, a number around 10,000 is appropriate. With PC 8801, when things went well, one iteration took about 3 min 25 s.

Discussion

Generally d^2 converges relatively fast but the convergence of the various parameters is slow. The reason seems to be that in the vicinity of the solution, the value of the deviation-inducing functions of the various parameters came close to 0 and this is a common characteristic seen in the various kinds of optimization methods. Consequently it is preferable not to put excessive faith in the calculated values of the parameters (Ruckdeschel 1982).

The method seeking the regression curve by the least square method is explained in detail by Nakagawa and Koyanagi (1982). For this, complex large programs must be prepared for use by large computers, as in the presented SALS. However, as program to be used with small computers, this level of program seems to be appropriate. With regard to data not suited to this program, it is preferable to proceed with large computers. It also appears that the method determining by integrals the residual functions as proposed by Shimazu (1979) takes too much time with small computers. p. 60

This program may be applied to most regression curves. It is also possible to find the partial differences by difference approximation and to make the program universally usable. When, however, one considers things such as the fact that the convergence becomes less stable because of the errors, it seems preferable to make a new preparation each time.

At the close of this article I want to thank Kiyohide ISHIOKA, chief researcher at the marine resources department of the Southwest Sea Regional Fisheries Laboratory as well as Fumihiko KATO, chief researcher at the resources department at the Sea of Japan Regional Fisheries Laboratory for their encouragement and advice.



第1図 変数の説明
Fig. 1. Illustration of variables.



第2図 パラメータ数が2の場合の Gauss-Seidel 法の収束

Fig. 2. Convergence of Gauss-Seidel method in the case of 2 parameters.



第3図 d^2 の収束状態

GS : Gauss-Seidel 法　GN : Gauss-Newton 法　データ：キダイの例
初期値：Gauss-Seidel 法で5回反復計算

Fig. 3. Convergence of d^2.
GS: Gauss-Seidel method. GN: Gauss-Newton method. Data: the example of porgy. Initial values: after 5 interations of Gauss-Seidel method.



第4図 パラメータ数が2の場合の最急降下法の収束

Fig. 4. Convergence of steepest descent method in the case of 2 parameters.



第5図 パラメータのスケーリング
g ：最急降下ベクトル
g' ：スケーリングしたパラメータの最急降下ベクトル

Fig. 5. Scaling of parameters.
g: steepest descent vector.
g': steepest descent vector of scaled parameter.

Table 1. Convergence of d^2.

Number of iterations	Marqualdt's method		Gauss-Seidel method	Steepest descent method
	$\lambda=1000$	$\lambda=10^4$		
0	798925	798925	798925	798925
1	30880	21717	95838	638443
2	6784	9405	25020	579662
3	6282	7646	11443	520681
4	6251	6810	8309	492315
5	6250	6530	7364	465906
6	6250	6453	6987	440723
7		6401	6802	417380
8		6359	6695	395116
9		6329	6626	374430
10		6300	6577	354749
11		6273	6539	336417
12		6257	6508	319014
13		6251	6481	302771
14		6250	6458	287377
15		6250	6439	272987
16			6421	259367
17			6406	246620
18			6392	234566
19			6380	223274
20			6370	212604

プログラムリスト（キダイのデータの入力例）

Program list (DATA: the example of porgy)

```
10    REM
20    REM POLYMODAL 3
30    REM
40    REM ﾆｭｳﾘｮｸ
50    READ NND,MCM,NCL,CWD,NIT,LAMBDA,NU
60    PRINT "ｾｲｷﾌﾞﾝﾌﾟ ﾉ ｶｽﾞ=";NND
70    PRINT "ｻｲｼｮｳ ﾉ ｶｲｷｭｳﾁ=";MCM
80    PRINT "ｶｲｷｭｳ ﾉ ｶｽﾞ    =";NCL
90    PRINT "ｶｲｷｭｳ ﾉ ﾊﾊﾞ    =";CWD
100   PRINT "     ﾊﾝﾌﾟｸｽｳ    =";NIT
110   PRINT "     LAMBDA     =";LAMBDA
120   PRINT "       NU       =";NU
130   PRINT
140   STOP
150   N3=3*NND
160   DIM F(NCL),X(NCL),DX(NCL),BIBUN(N3),ND(NND,NCL)
170   DIM HENSU(N3),KEISU(N3,N3),TEISU(N3),ZOBUN(N3),HENSU2(N3)
180   FOR K=1 TO NCL
190      X(K)=MCM+(K-1)*CWD
200      READ F(K)
210      PRINT "F(";X(K);")=";F(K)
220   NEXT K
230   PRINT
240   STOP
250   FOR I=1 TO NND
260      S3=I:S2=S3+NND:S1=S2+NND
270      READ HENSU(S1),HENSU(S2),HENSU(S3)
280      PRINT "I=";I
290      PRINT "ﾒﾝﾔｷ=";HENSU(S1),"ﾍｲｷﾝ=";HENSU(S2),"ﾌﾞﾝｻﾝ=";HENSU(S3)
300   NEXT I
310   STOP
500   REM ｹｲｻﾝ
510   P9=.398942:P8=-.5
520   FOR I=1 TO N3
530      HENSU2(I)=HENSU(I)
540   NEXT I
550   GOSUB *CLD2
560   D2=D3
570   PRINT "D2=";D2
580   PRINT
590   FOR KK1=1 TO NIT+1
600      REM ｼｮｷｶ
610      FOR I=1 TO N3
620         TEISU(I)=0
630         FOR J=I TO N3
640            KEISU(I,J)=0
650      NEXT J:NEXT I
660      REM ｿｳｸ
670      FOR K=1 TO NCL
680         FOR I=1 TO NND
690            S3=I:S2=S3+NND:S1=S2+NND
700            P1=HENSU(S1):P2=X(K)-HENSU(S2):P3=HENSU(S3)
710            P6=ND(I,K)
720            BIBUN(S1)=P6/P1*HENSU(S1)
730            BIBUN(S2)=P6*P2/P3/P3*HENSU(S2)
740            BIBUN(S3)=P6*(P2*P2-P3*P3)/P3/P3/P3*HENSU(S3)
750         NEXT I
760         D1=DX(K)
770         REM ｹｲｽｳ
780         FOR I=1 TO N3
790            TEISU(I)=TEISU(I)+D1*BIBUN(I)
800            FOR J=I TO N3
810               KEISU(I,J)=KEISU(I,J)+BIBUN(I)*BIBUN(J)
820      NEXT J:NEXT I:NEXT K
830      LAMBDA2=0
840      K2=0
850     *REP
860      K2=K2+1
870      IF K2>11 GOTO *OWARI
880      PRINT
890      PRINT "LAMBDA=";LAMBDA
900      PRINT
910      FOR I=1 TO N3
920         KEISU(I,I)=KEISU(I,I)+LAMBDA-LAMBDA2
930      NEXT I
940      REM ｾﾞﾝｼﾝ ｼｮｳｷｮ
950      FOR I=1 TO N3-1
960         FOR K=I+1 TO N3
970            Q1=KEISU(I,K)/KEISU(I,I)
980            TEISU(K)=TEISU(K)-Q1*TEISU(I)
990            FOR J=K TO N3
1000              KEISU(K,J)=KEISU(K,J)-Q1*KEISU(I,J)
1010     NEXT J:NEXT K:NEXT I
1020     REM ｺｳﾀｲ ﾀﾞｲﾆｭｳ
1030     ZOBUN(N3)=TEISU(N3)/KEISU(N3,N3)
1040     FOR I=N3-1 TO 1 STEP -1
1050        T1=TEISU(I)
1060        FOR J=I+1 TO N3
1070           T1=T1-ZOBUN(J)*KEISU(I,J)
1080        NEXT J
1090        ZOBUN(I)=T1/KEISU(I,I)
1100     NEXT I
1110     FOR I=1 TO N3
1120        ZOBUN(I)=ZOBUN(I)*HENSU(I)
1130        HENSU2(I)=HENSU(I)+ZOBUN(I)
1140     NEXT I
1150     GOSUB *CLD2
1160     IF D3>=D2 THEN LAMBDA2=LAMBDA:LAMBDA=LAMBDA*NU:GOTO *REP
1170     REM ｼｭｳｾｲ
1180     LAMBDA=LAMBDA/NU
1190     D2=D3
1200     FOR I=1 TD N3
1210        HENSU(I)=HENSU2(I)
1220     NEXT I
1230     GOSUB *SHUTU
1240 NEXT KK1
1250 *OWARI
1260  END
1500 *CLD2
1510  D3=0
1520  FOR K=1 TO NCL
1530     F1=0
1540     FOR I=1 TO NND
1550        S3=I:S2=S3+NND:S1=S2+NND
1560        P1=HENSU2(S1):P2=X(K)-HENSU2(S2):P3=HENSU2(S3)
1570        ND(I,K)=P9*P1/P3*EXP(P8*P2*P2/P3/P3)
1580        F1=F1+ND(I,K)
1590     NEXT I
1600     D1=F(K)-F1
1610     DX(K)=D1
1620     D3=D3+D1*D1
1630  NEXT K
1640 RETURN
2000 *SHUTU
```

```
2010  REM ｼｭﾂﾘｮｸ
2020  PRINT "D2=";D2
2030  PRINT
2040  PRINT "ﾊﾝﾌﾟｸｽｳ=";KK1
2050  FOR I=1 TO NND
2060     PRINT "I=";I
2070     S3=I:S2=S3+NND:S1=S2+NND
2080     PRINT "ﾒﾝｾｷ=";HENSU(S1),"ﾍｲｷﾝ=";HENSU(S2),"ﾌﾞﾝｻﾝ=";HENSU(S3)
2090  NEXT I
2100 RETURN
3000 DATA 5,7.5,29,1,20,10000,2
3010 DATA 7,79,509,2240,2341,623,476,1230,1439,921,448,512,719,673
3020 DATA 445,341,310,228,168,140,114,64,22,0,2,2,0,0,1
3030 DATA 5000,11,1,4000,15.5,1,3000,20,1.5,1000,24,1.5,500,27,1.5
```

変数の説明

Correspondence of variables

NND	: number of normal distributions	D3	: d^2_{n+1}
MCM	: minimum class mark		
NCL	: number of classes	HENSU(I)	: $\alpha_{i,n}$
CWD	: class width	HENSU 2 (I)	: $\alpha_{i,n+1}$
NIT	: number of iterations	BIBUN (I)	: $\dfrac{\partial f}{\partial \alpha_i}$
F(K)	: F	KEISU (I, J)	: $\sum\limits_x \dfrac{\partial f}{\partial \alpha_i}\ \dfrac{\partial f}{\partial \alpha_j}$
X(K)	: x		
DX(K)	: d_x	TEISU (I)	: $\sum\limits_x d_x \dfrac{\partial f}{\partial \alpha_i}$
		ZOBUN (I)	: $\Delta\alpha_i$
D2	: d^2_n	ND(J, K)	: $K_i \cdot N(\mu_i, \sigma_i, x)$

Bibliography

1、 赤嶺達郎 (1982). Polymodal な度数分布を正規分布へ分解する BASIC プログラム. 日水研報告(33)：163-166.

2. 中川 徹・小柳義夫 (1982). 最小二乗法による実験データ解析―プログラム **SALS**. 東京大学出版会, 東京, 206 pp.

3, Ruckdeschel, F. R. (1982). 科学計算のための BASIC サブルーチン集 2 (下) (長谷川勝也, 石原辰雄訳). 現代数学社, 東京, 245 pp.

4, 嶋津靖彦 (1979). 体長組成から年齢組成を推定する一方法. 昭和 54 年度漁業資源研究会議, 西日本底魚部会会議報告：36-48.

5、 田中昌一 (1956). Polymodal な度数分布の一つの取扱方及びそのキダイ体長組成解析への応用. 東海水研報告(14)：1-13.

6、 戸川隼人 (1971). マトリクスの数値計算. オーム社, 東京, 323 pp.

1. This author (1982). Polymodal na Dosu Bunpu o Seiki Bunpu e Bunkai suru BASIC puroguramu (BASIC program to analyse the polymodal frequency distribution into normal distribution) this journal (33): 163-166.

2. Toru NAKAGAWA, Yoshio KOYANAGI (1982). Saisho Nijoho ni yoru Jikken Deta Kaiseki puroguramu SALS (The SALS program for the analysis of experimental data based on the least square method) Publ. Soc. of Tokyo University, Tokyo, 206 pp.

3. Ruckdeschel, F. R. (1982). Kagaku Keisan no tame no BASIC Sabu-ruchinu-shu 2 (Collected BASIC sub-routines for scientific calculations 2) (2nd part) (translated by K. Hasegawa) Gendai Sugaku Co., Tokyo, 245 pp.

4. Yasuhiko SHIMAZU (1979). Taicho Sosei kara Nenrei Sosei o Suitei suru hito-Hoho (One method to estimate the age composition from the body length composition). Showa 54 nendo Gyogyo Shigen Kenkyu Kaigi (1979 study meeting for Fisheries Resources) Nishi Nihon Teigyo-bukai Kaigi Hokoku (Report from the Western Japan Bottom Fish Sectional Group) 36-48.

5. Shoichi TANAKA (1956). Polymodal na Dosu Bunpu no hitotsu no Toriatsukaikata oyobi sono Kidai Taicho Sosei Bunseki e no Oyo (One method to handle polymodal frequency distributions and application to the analysis of porgy body length composition). Tokai Suiken Hokoku (Abbreviated name; report from the Eastern Seas Fisheries Research Center) (14): 1-13.

6. Hayato TOGAWA (1971). Matorikusu no Suchi Keisan (Matrix numerical values computation). Omu Co. Tokyo, 323 pp.